#-------------------------------------------------
#
# Project created by QtCreator 2012-04-30T15:08:40
#
#-------------------------------------------------

QT       += core gui

TARGET = cs235app
TEMPLATE = app


SOURCES += main.cpp\
    data_plot.cpp \
    print_labeled_spinbox.cpp \
    print_labels.cpp \
    slider_with_box.cpp \
    sliders.cpp \
    phidget_servo_motor.cpp \
    my_DC_motor.cpp \
    my_servo_motor.cpp \
    phidget_DC_motor.cpp \
    mainwindow.cpp \
    dialog.cpp \
    scarasolver-demo.cpp

HEADERS  += \
    data_plot.h \
    print_labeled_spinbox.h \
    print_labels.h \
    slider_with_box.h \
    sliders.h \
    phidget_servo_motor.h \
    my_DC_motor.h \
    my_servo_motor.h \
    phidget_DC_motor.h \
    mainwindow.h \
    dialog.h \
    scarasolver.h

FORMS    += \
    dialog.ui


###################################################################
# Link with phidgets (install the installation package separately)
###################################################################
win32 {
  LIBS += -L"C:/Program Files/Phidgets/x86" -lphidget21 
  INCLUDEPATH += "C:/Program Files/Phidgets"
}
else {
  LIBS += -lphidget21
}

###################################################################


###################################################################
#
# Include the following to link with Qwt.
# /Jonas
#
###################################################################
# Qwt Widget Library
# Copyright (C) 1997   Josef Wilgen
# Copyright (C) 2002   Uwe Rathmann
#
# This library is free software; you can redistribute it and/or
# modify it under the terms of the Qwt License, Version 1.0
###################################################################
QWT_ROOT = $${PWD}/../qwt/
include( $${QWT_ROOT}/qwtconfig.pri )
include( $${QWT_ROOT}/qwtbuild.pri )

INCLUDEPATH += $${QWT_ROOT}/src
DEPENDPATH  += $${QWT_ROOT}/src
DESTDIR      = $${QWT_ROOT}/examples/bin

QMAKE_RPATHDIR *= $${QWT_ROOT}/lib

contains(QWT_CONFIG, QwtFramework) {

    LIBS      += -F$${QWT_ROOT}/lib
}
else {

    LIBS      += -L$${QWT_ROOT}/lib
}

IPATH       = $${INCLUDEPATH}
qtAddLibrary(qwt)
INCLUDEPATH = $${IPATH}

contains(QWT_CONFIG, QwtSvg) {

    QT += svg
}
else {

    DEFINES += QWT_NO_SVG
}


win32 {
    contains(QWT_CONFIG, QwtDll) {
        DEFINES    += QT_DLL QWT_DLL
    }
}
###################################################################































